GMQL exercises
GMQL for the analysis of (epi)genomic data features


Ex. 1 –	Find the genomic regions of human genome GRCh38 covered by signal peaks of both H3K4me1 and H3K27ac histone marks, which are potential active enhancers, in the HCT116 human cell line. Use ENCODE narrowpeak data (dataset of November 2017) regarding experiment target H3K4me1 and H3K27ac.
Then, count the number of regions in the input data and in the result. 
Report obtained sample and region numbers, and discuss the results.

Version 1 

#1 Loading the data
H3K4me1 = SELECT(biosample_term_name == 'HCT116' AND experiment_target == 'H3K4me1-human') GRCh38_ENCODE_NARROW_NOV_2017;
H3K27ac = SELECT(biosample_term_name == 'HCT116' AND experiment_target == 'H3K27ac-human') GRCh38_ENCODE_NARROW_NOV_2017;

#2 Combining multiple replicates in a single sample
H3K4me1_1 = COVER(1, ANY) H3K4me1;
H3K27ac_1 = COVER(1, ANY) H3K27ac;

#3 Generating single input dataset
K4me1_K27ac = UNION() H3K4me1_1 H3K27ac_1;

#4 Extracting the portion of intersecting regions
RES_0 = COVER(2, ANY) K4me1_K27ac;

#5 Counting output regions and saving the result
RES = EXTEND(_RegionCountOutput AS COUNT()) RES_0;
MATERIALIZE RES INTO res;
    
#6 Counting and saving input regions
HM_0 = UNION() H3K4me1 H3K27ac;
HM = EXTEND(_RegionCountInput AS COUNT()) HM_0;
MATERIALIZE HM INTO HM;
 
Execution time: 	00:03:53 
RES samples: 		1 
RES regions: 		93 562 
HM samples: 		16
HM regions: 		1 490 710

Version 2 

#1 Loading the data
H3K4me1 = SELECT(biosample_term_name == 'HCT116' AND experiment_target == 'H3K4me1-human') GRCh38_ENCODE_NARROW_NOV_2017;
H3K27ac = SELECT(biosample_term_name == 'HCT116' AND experiment_target == 'H3K27ac-human') GRCh38_ENCODE_NARROW_NOV_2017;

#2 Combining multiple replicates in a single sample
H3K4me1_1 = COVER(1, ANY) H3K4me1;
H3K27ac_1 = COVER(1, ANY) H3K27ac;

#3 Extracting the portion of intersecting regions
RES_0 = JOIN(distance < 0; output: INT) H3K4me1_1 H3K27ac_1;

#4 Counting output regions and saving the result
RES = EXTEND(_RegionCountOutput AS COUNT()) RES_0;
MATERIALIZE RES INTO res;

#5 Counting and saving input regions
HM_0 = UNION() H3K4me1 H3K27ac;
HM = EXTEND(_RegionCountInput AS COUNT()) HM_0;
MATERIALIZE HM INTO HM;

Execution time: 	00:03:50 (00:04:33 without initial covers)
RES samples: 		1 (64 without initial covers)
RES regions: 		93 562 (3 326 076 without initial covers)
HM samples: 		16
HM regions: 		1 490 710


Ex. 2 – 	Identify HG19 DNA regions bound by the JUN transcription factor that overlap binding sites of FOS transcription factor, which could indicate co-binding events and provide information for finding functionally related factors. Use ENCODE narrowpeak data (dataset of November 2017) regarding experiment target JUN and FOS in the HeLa-S3 cell line; particularly, use only the single sample data produced by the lab 'ENCODE Consortium Analysis Working Group'.
Then, count the number of regions in the input data and in the result. 
Report obtained sample and region numbers, and discuss the results.

#1 Loading the data
FOS = SELECT(biosample_term_name == 'HeLa-S3' AND experiment_target == 'FOS-human' AND lab == 'ENCODE Consortium Analysis Working Group') HG19_ENCODE_NARROW_NOV_2017;
JUN = SELECT(biosample_term_name == 'HeLa-S3' AND experiment_target == 'JUN-human' AND lab == 'ENCODE Consortium Analysis Working Group') HG19_ENCODE_NARROW_NOV_2017;

#2 Identifying the JUN binding regions overlapping with FOS binding regions
RES_0 = JOIN(distance < 0; output: LEFT_DISTINCT) FOS JUN;

#3 Counting output regions and saving the result 
RES = EXTEND(_RegionCountOutput AS COUNT()) RES_0;
MATERIALIZE RES INTO res;

#4 Counting and saving input regions
TF_0 = UNION() FOS JUN;
TF = EXTEND(_RegionCountInput AS COUNT()) TF_0;
MATERIALIZE TF INTO TF;

Execution time: 	00:02:06
RES samples: 		1
RES regions: 		7692
TF samples: 		2
TF regions: 		31 228


Ex. 3 - 	Identify DNA promotorial regions bound by the MYC transcription factor that present somatic mutations (if any) in breast cancer patients. Use narrowpeak data from the ChIP-seq assay of MYC binding sites for cell line MCF-7 from ENCODE (HG19 dataset of November 2016), DNA-seq data from TCGA breast cancer (BRCA tumor tag) patients with clinical followup reason 'additional new tumor event', and genomic region annotations data (promoter, TSS, gene) from RefSeq (dataset HG19_BED_ANNOTATION). 
Then, for each selected patient count the number of MYC bound promoters with patent’s mutations. 
Report obtained sample and region numbers, and discuss the results.

#1 Loading data
MYC = SELECT(assay == 'ChIP-seq' AND biosample_term_name == 'MCF-7' AND 
experiment_target == 'MYC-human') HG19_ENCODE_NARROW_NOV_2016;

BRCA = SELECT(manually_curated__tumor_tag == 'brca' AND clinical_follow_up__followup_reason == 'additional new tumor event') HG19_TCGA_dnaseq;

# PROM = SELECT(annotation_type == 'promoter') HG19_BED_ANNOTATION;
GENE = SELECT(provider == 'RefSeq' AND annotation_type == 'gene') HG19_BED_ANNOTATION;
# PROM = PROJECT(region_update: start AS start - 2000, stop AS start + 1000) GENE;
#MATERIALIZE GENE INTO GENE;
# In the input GENE dataset for the same gene there can be multiple regions with 
# different gene IDs.
# So, calculating directly the promoter regions I can obtain multiple replicate
# regions for the same promoter.
# To avoid it, I can perform the following operations 
TSS = PROJECT(region_update: stop AS start + 1) GENE;
TSS_1 = COVER(1, ANY; aggregate: BAG(name)) TSS;
PROM = PROJECT(region_update: start AS start - 2000, stop AS start + 1000) TSS_1;

#2 Combining MYC replicates 
MYC_1 = COVER(1, ANY) MYC;

#3 Extracting MYC-binding promoters
# PROM_MYC = JOIN(distance < 0; output: left_distinct) PROM MYC_1;

PROM_MYC_0 = MAP() PROM MYC_1;
PROM_MYC = SELECT(region: count_PROM_MYC_1 > 0) PROM_MYC_0;
#MATERIALIZE PROM_MYC INTO PROM_MYC;

#4 Extracting BRCA-mutated MYC-binding promoters
# PROM_MYC= MERGE() PROM_MYC_0; 
# Using here MERGE instead of COVER before is not the proper solution since it 
# maintains several replicated regions in the MERGE output coming from the 
# different MERGE input samples 

PROM_MYC_BRCA = JOIN(distance < 0; output: left_distinct) PROM_MYC BRCA;
#PROM_MYC_BRCA_0 = MAP() PROM_MYC BRCA;
#PROM_MYC_BRCA = SELECT(region: count_PROM_MYC_BRCA > 0) PROM_MYC_BRCA_0;

#5 Counting BRCA-mutated MYC-binding promoters for each patient
RES = EXTEND(_prom_count AS COUNT()) PROM_MYC_BRCA;
MATERIALIZE RES INTO RES;

Execution time:	00:03:23	
RES samples: 		3
RES regions: 		6


Ex. 4 –	Differential binding search. Answer the following questions using HG19 ENCODE narrowpeak data with output type ‘optimal idr thresholded peaks’ regarding JUN experiment target from the ChIP-seq assay (dataset of November 2017) for cell line k562 (chronic myelogenous leukemia), with 1 control and 2 cases treated using interferon alpha or gamma, respectively, each of the latter ones with two replicas. Comment required GMQL statement parameters not specified in the exercise text.
Report obtained sample and region numbers, and discuss the results.

Question 1: Combining replicate samples (if existing) by considering only regions in at least two replicas and calculating the minimum pvalue of the combined regions, find those JUN binding regions in each treated cases where there are no JUN binding regions in the control case.

# Select the control
JUN = SELECT(assay == 'ChIP-seq' AND experiment_target == 'JUN-human' AND 
biosample_term_name == 'K562' AND 
output_type == 'optimal idr thresholded peaks' AND 
NOT (biosample_treatments == '*')) HG19_ENCODE_NARROW_NOV_2017;
		
# Only one control sample is selected with the used selection parameters
#JUN = COVER(2, ANY; aggregate: pValue AS MIN(pvalue)) JUN_2;

# Select treated samples
JUN_IFNa_2 = SELECT(assay == 'ChIP-seq' AND experiment_target == 'JUN-human' AND biosample_term_name == 'K562' 
AND output_type == 'optimal idr thresholded peaks' 
AND biosample_treatments == 'interferon alpha') HG19_ENCODE_NARROW_NOV_2017;
JUN_IFNg_2 = SELECT(assay == 'ChIP-seq' AND experiment_target == 'JUN-human' AND biosample_term_name == 'K562' 
AND output_type == 'optimal idr thresholded peaks' 
AND biosample_treatments == 'interferon gamma') HG19_ENCODE_NARROW_NOV_2017;

# Extract the treated regions in at least two replicas and their minimum p-value
JUN_IFNa_3 = COVER(2, ANY; aggregate: min_pValue AS MIN(pvalue)) JUN_IFNa_2;
JUN_IFNg_3 = COVER(2, ANY; aggregate: min_pValue AS MIN(pvalue)) JUN_IFNg_2;

# Extract treated regions which do not overlap with control ones (Question 1)
#JUN_IFNa = JOIN(DGE(0), MD(100); output: left_distinct) JUN_IFNa_3 JUN;
#JUN_IFNg = JOIN(DGE(0), MD(100); output: left_distinct) JUN_IFNg_3 JUN;
# MD() condition is needed to have a well-formed genometric clause
# Subjective value 100 can influence final results
# The JOIN(DGE(0), MD(100); ...) solution is wrong since for each reference region
# the extracted regions could overlap other reference regions ....
JUN_IFNa = DIFFERENCE() JUN_IFNa_3 JUN;
JUN_IFNg = DIFFERENCE() JUN_IFNg_3 JUN;

MATERIALIZE JUN INTO JUN;
MATERIALIZE JUN_IFNa INTO JUN_IFNa;
MATERIALIZE JUN_IFNg INTO JUN_IFNg;

Execution time:			00:02:54
JUN samples: 			1
JUN regions: 				9 848
JUN_IFNa samples: 			1
JUN_IFNa regions: 			454	(2 979 with JOIN(DGE(0), MD(100); ...))
JUN_IFNg samples: 			1
JUN_IFNg regions: 			1 121	(4 002 with JOIN(DGE(0), MD(100); ...))


Question 2: Find the JUN binding regions that are differentially present between treated cases (e.g., in alpha and not in gamma interferon treated samples) and not in control.

# All treated samples: union of all treated regions non-overlapping with control ones
JUN_IFNag = UNION() JUN_IFNa JUN_IFNg;

# Regions that are present in treatment alpha and not in treatment
# gamma (Question 2)
# A possible solution:
#JUN_IFNag_ONLYa_0 = MAP() JUN_IFNa JUN_IFNg;
#JUN_IFNag_ONLYa = SELECT(region: count_ JUN_IFNa_ JUN_IFNg == 0) JUN_IFNag_ONLYa_0;
# or better:
JUN_IFNag_ONLYa = DIFFERENCE() JUN_IFNa JUN_IFNg;

MATERIALIZE JUN_IFNag_ONLYa INTO JUN_IFNag_ONLYa;

# Regions that are present in treatment gamma and not in treatment
# alpha (Question 2)
JUN_IFNag_ONLYg = DIFFERENCE() JUN_IFNg JUN_IFNa;

MATERIALIZE JUN_IFNag_ONLYg INTO JUN_IFNag_ONLYg;

Execution time:			00:02:43 (cumulated)
JUN_IFNag_ONLYa samples: 	1
JUN_IFNg_ONLYa regions: 	258	(729 with JOIN(DGE(0), MD(100); ...))
JUN_IFNag_ONLYg samples: 	1
JUN_IFNg_ONLYg regions: 	925	(1 755 with JOIN(DGE(0), MD(100); ...))


Question 3: Find the RefSeq genes (dataset HG19_BED_ANNOTATION) that overlap with JUN binding regions either in control or treated samples.

# Extract all genes
GENE = SELECT(annotation_type == 'gene' AND provider == 'RefSeq') HG19_BED_ANNOTATION;

# All samples: union of regions in controls or in treated samples
JUN_ALL_0 = UNION() JUN JUN_IFNag;

# Combine all JUN control or treated regions in a single sample
JUN_ALL = COVER(1, ANY) JUN_ALL_0;

# Extract genes intersecting with regions in control or treated samples (Question 3)
#GENE_JUN_ALL = JOIN(distance < 0; output: left_distinct) GENE JUN_ALL;
GENE_JUN_ALL_0 = MAP() GENE JUN_ALL;
GENE_JUN_ALL = SELECT(count_GENE_JUN_ALL > 0) GENE_JUN_ALL_0;

MATERIALIZE GENE_JUN_ALL INTO GENE_JUN_ALL;

Execution time:			00:02:44
GENE_JUN_ALL samples: 		0
GENE_JUN_ALL regions: 		0


Question 4: In each interferon alpha and gamma treated and control condition, find the overall average JUN ChIP-seq signal in gene promoters (areas surrounding the TSS) intersecting JUN binding regions either in control or treated samples.

